
	Reason for change:
	Two issues have been observed related to TCP connection handling in IP PTC.

1- When the TTCN closes a TCP connection, there exists the possibility that the UE may asynchronously attempt  to close the same connection at about the same time. This can lead to a crossover case where after sending a “TCP close” request to the system adapter, the TTCN may recieve a “TCP Close” indication before the “TCP CloseCnf” response. In this case the first indication will never be handled by the TTCN and will remain in the message queue, meaning that the “CloseCnf” will never be received and the test case execution stalls.

2- On processing a SIP response to a request received over TCP, the IP PTC will attempt to send the response over the same TCP connection. If the connection no longer exists (i.e. a “TCP close” indication has been received from the system adapter while the request is being processed) the current TTCN implementation throws a fatal error “no valid TCP connection”.
It has been seen that some UEs performing IMS deregistration during power off will send a SIP REGISTER over TCP and then immediately close the TCP connection that was used, without waiting for a 200 OK response. This behaviour can cause the TTCN to assign an ERROR verdict during test case postamble.



	
	

	Summary of change:
	1 - Add an alt block in function fl_TCP_Close() so that a “Close” indication is handled if received while waiting for “CloseCnf”. No processing is needed for this indication as the socket is already being closed.

2  - Remove the fatal error thrown by IP PTC in the case where a SIP downink message is required to be sent over a TCP connection which no longer exists. In this case the message is discarded without sending.
Note : According to RFC 3261 18.2.2, this is a valid scenario but should be handled differently by a SIP server. The RFC states :
“If the "sent-protocol" is a reliable transport protocol such as

TCP or SCTP, or TLS over those, the response MUST be sent using

the existing connection to the source of the original request

that created the transaction, if that connection is still open.

….

If that connection is no longer open, the server SHOULD open a

connection to the IP address in the "received" parameter, if

present, using the port in the "sent-by" value, or the default

port for that transport, if no port is specified. 

…“

Since the behaviour (closing a TCP connection immediately after sending a SIP request without waiting for a response) has currently only been observed by a UE in the process of powering down, it is proposed to discard the message rather than following the steps detailed in the RFC to open a new TCP connection on which to send the response.



	
	

	Consequences if not approved:
	A conformant UE can fail IMS test cases when TCP transport is used.

Change 1
	Function name
	fl_TCP_Close()

	Reason for change
	When the TTCN closes a TCP connection, there exists the possibility that the UE may attempt  to close the same connection at about the same time. This may lead to a crossover case where after sending a “TCP close” request to the system adapter, the TTCN may recieve a “TCP Close” indication to TTCN before the “TCP CloseCnf” response. In this case the first indication will never be handled by the TTCN and will remain in the message queue, meaning that the “CloseCnf” will never be received and the test case execution stalls.



	Summary of change
	Add an alt block in function fl_TCP_Close() so that a “Close” indication is handled if received while waiting for “CloseCnf”. No processing is needed for this indication as the socket is already being closed.



	TTCN module
	TCP_Functions.ttcn

	MCC160 Comment
	Accepted


Before change

  function fl_TCP_Close(template (value) IP_Connection_Type p_IP_Connection,

                        boolean p_WaitForCNF := true) runs on IP_PTC

  {

    IP_SOCK.send(cs_TCP_CLOSE_REQ(p_IP_Connection));

    if (p_WaitForCNF) {

      IP_SOCK.receive(cr_TCP_CLOSE_CNF(p_IP_Connection));

    }

  }
After change

    function fl_TCP_Close(template (value) IP_Connection_Type p_IP_Connection,

                        boolean p_WaitForCNF := true) runs on IP_PTC

  {

    IP_SOCK.send(cs_TCP_CLOSE_REQ(p_IP_Connection));

    if (p_WaitForCNF) {

    alt{

      [] IP_SOCK.receive(cr_TCP_CLOSE_CNF(p_IP_Connection)) { }
      [] IP_SOCK.receive(cr_TCP_CLOSE_IND(p_IP_Connection)) { repeat; }

      }
    }

  }
Change 2
	Function name
	fl_IP_IMS_Server_SendDL()

	Reason for change
	On processing a SIP response to a request received over TCP, the IP PTC will attempt to send the response over the same TCP connection. If the connection no longer exists (i.e. a “TCP close” indication has been received from the system adapter while the request is being processed) the current TTCN implementation throws a fatal error “no valid TCP connection”.

It has been seen that some UEs during power off will send a SIP REGISTER over TCP and then immediately close the TCP connection that was used, without waiting for a 200 OK response. This behaviour can cause the TTCN to assign an ERROR verdict during test case postamble.

	Summary of change
	Remove the fatal error thrown by IP PTC in the case where a SIP downink message is required to be sent over a TCP connection which no longer exists. In this case the message is discarded without sending

	TTCN module
	TCP_Functions.ttcn

	MCC160 Comment
	Accepted (but using “else” rather than “return”)


Before change

   function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,

                                   IMS_RoutingInfo_Type p_RoutingInfo,

                                   IMS_RequestResponse_Type p_RequestResponse,

                                   bitstring p_Data) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_SecurityContextEnum_Type v_SecurityContextEnum;

    var IMS_SecurityContext_Type v_SecurityContext;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var IP_Connection_Type v_NewTcpConnectionByRef;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_IsUnprotected;

    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_SecurityContextEnum := p_RoutingInfo.Security;

    } else {

      if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {         /* the latest security context is preferred */

        v_SecurityContextEnum := protectedContext2;

      } else if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

        v_SecurityContextEnum := protectedContext1;

      } else {

        v_SecurityContextEnum := unprotected;

      }

    }

    if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, v_SecurityContextEnum)) {

      FatalError(__FILE__, __LINE__, "IMS security not established");

    }

    v_SecurityContext := valueof(fl_IMS_Server_GetSecurityContext(p_ImsServer, v_SecurityContextEnum));

    v_IsUnprotected := (v_SecurityContextEnum == unprotected);

    v_LocalRemotePort := fl_IP_IMS_GetPortsTX(v_SecurityContext, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;

    } else if (fl_ImsServer_UeIsRegistered(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (p_RequestResponse) {

          case (response) {

            v_TCP_Connection := v_SecurityContext.TcpServer;

          }

          case (request) {

            v_TCP_Connection := v_SecurityContext.TcpClient;

            if (not isvalue(v_TCP_Connection)) {

              if (v_IsUnprotected) {

                v_LocalPort := omit;  // in case of an unprotected TCP connection to UE there is no need to specify any specific local port but we can use an ephemeral port instead

              }

              v_NewTcpConnectionByRef := f_TcpClient_ConnectReq(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), ims);

              alt {

                [] a_IP_IMS_TCP_ConnectionControl(p_ImsServer) {  /* @sic R5s140713 - MCC160 implementation sic@ */

                  repeat;

                }

                [] a_TcpClient_ConnectCnf(v_NewTcpConnectionByRef, p_ImsServer.DrbInfo) {

                  v_SecurityContext.TcpClient := v_NewTcpConnectionByRef;

                  fl_IMS_Server_SetSecurityContext(p_ImsServer, v_SecurityContextEnum, v_SecurityContext); // @sic R5s141322 change 2 sic@

                  v_TCP_Connection := v_NewTcpConnectionByRef;

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          FatalError(__FILE__, __LINE__, "no valid TCP connection");
        }

        IP_SOCK.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }
After change

function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,

                                   IMS_RoutingInfo_Type p_RoutingInfo,

                                   IMS_RequestResponse_Type p_RequestResponse,

                                   bitstring p_Data) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_SecurityContextEnum_Type v_SecurityContextEnum;

    var IMS_SecurityContext_Type v_SecurityContext;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var IP_Connection_Type v_NewTcpConnectionByRef;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_IsUnprotected;

    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_SecurityContextEnum := p_RoutingInfo.Security;

    } else {

      if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {         /* the latest security context is preferred */

        v_SecurityContextEnum := protectedContext2;

      } else if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

        v_SecurityContextEnum := protectedContext1;

      } else {

        v_SecurityContextEnum := unprotected;

      }

    }

    if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, v_SecurityContextEnum)) {

      FatalError(__FILE__, __LINE__, "IMS security not established");

    }

    v_SecurityContext := valueof(fl_IMS_Server_GetSecurityContext(p_ImsServer, v_SecurityContextEnum));

    v_IsUnprotected := (v_SecurityContextEnum == unprotected);

    v_LocalRemotePort := fl_IP_IMS_GetPortsTX(v_SecurityContext, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;

    } else if (fl_ImsServer_UeIsRegistered(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (p_RequestResponse) {

          case (response) {

            v_TCP_Connection := v_SecurityContext.TcpServer;

          }

          case (request) {

            v_TCP_Connection := v_SecurityContext.TcpClient;

            if (not isvalue(v_TCP_Connection)) {

              if (v_IsUnprotected) {

                v_LocalPort := omit;  // in case of an unprotected TCP connection to UE there is no need to specify any specific local port but we can use an ephemeral port instead

              }

              v_NewTcpConnectionByRef := f_TcpClient_ConnectReq(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), ims);

              alt {

                [] a_IP_IMS_TCP_ConnectionControl(p_ImsServer) {  /* @sic R5s140713 - MCC160 implementation sic@ */

                  repeat;

                }

                [] a_TcpClient_ConnectCnf(v_NewTcpConnectionByRef, p_ImsServer.DrbInfo) {

                  v_SecurityContext.TcpClient := v_NewTcpConnectionByRef;

                  fl_IMS_Server_SetSecurityContext(p_ImsServer, v_SecurityContextEnum, v_SecurityContext); // @sic R5s141322 change 2 sic@

                  v_TCP_Connection := v_NewTcpConnectionByRef;

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          //REMOVED FatalError(__FILE__, __LINE__, "no valid TCP connection"); 

          return;
        }

        IP_SOCK.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }
